/home/m262f99/KOEPF/worksheetsV.4/hw6ansm.mws

Math 262a, Fall 1999, Glenn Tesler
Homework 6

```
> read `hsum.mpl`;
```

Copyright 1998 Wolfram Koepf, Konrad-Zuse-Zentrum Berlin

Koepf # 9.1

```
> recpoly(n*s(n+1)-(n+20)*s(n),s(n));
```

$$\delta_0\, n\,(n+19)\,(n+18)\,(n+17)\,(n+16)\,(n+15)\,(n+14)\,(n+13)\,(n+12)\,(n+11)\,(n+10)\,(n+9)\,(n+8)\,(n+7)\,(n+6)\,(n+5)\,(n+4)\,(n+3)\,(n+2)\,(n+1)$$

```
> recpoly(n*s(n+1)-(n+40)*s(n),s(n));
```

$$\delta_0\, n\,(n+39)\,(n+38)\,(n+37)\,(n+36)\,(n+35)\,(n+34)\,(n+33)\,(n+32)\,(n+31)\,(n+30)\,(n+29)\,(n+28)\,(n+27)\,(n+26)\,(n+25)\,(n+24)\,(n+23)\,(n+22)\,(n+21)\,(n+20)\,(n+19)\,(n+18)\,(n+17)\,(n+16)\,(n+15)\,(n+14)\,(n+13)\,(n+12)\,(n+11)\,(n+10)\,(n+9)\,(n+8)\,(n+7)\,(n+6)\,(n+5)\,(n+4)\,(n+3)\,(n+2)\,(n+1)$$

Koepf # 9.3(a)

```
> rec:=
  sumrecursion(hyperterm([-n,a,a+1/2,b],[2*a,(b-n+1)/2,(b-
  n)/2+1],1,k),k,s(n));
```

$$rec := (n+2+b)\,(1+n-b)\,(-b+n)\,(n+2\,a+1)\,\mathrm{s}(n+2) - 2\,(n+1+b)\,(-b+n)\,(n+a+1)\,(1+2\,a+n-b)\,\mathrm{s}(n+1) + (n+1)\,(n+b)\,(1+2\,a+n-b)\,(2\,a-b+n)\,\mathrm{s}(n) = 0$$

```
> rec2hyper(rec,s(n));
```

$$\left\{\frac{(n+1)\,(n+b)\,(2\,a-b+n)}{(n+1+b)\,(-b+n)\,(n+2\,a)},\ \frac{(n+b)\,(2\,a-b+n)}{(n+1+b)\,(-b+n)}\right\}$$

Koepf # 9.12

```
> rec:=sumrecursion(
  (-1)^k*binomial(r-s-k,k)*binomial(r-2*k,n-k)/(r-n-k+1),k
  ,s(n));
```

$$rec := (n+2)\,(2\,n-r+3)\,(n+1-r+s)\,\mathrm{s}(n+2)$$

$$+ (2n+1-r)(2n^2+2n-2rn+sr-s^2-2r)\,\mathrm{s}(n+1)$$
$$+ (2n-r-1)(-r+n-1)(n-s)\,\mathrm{s}(n) = 0$$

```
> rechyper(rec,s(n));
```

$$\{-\frac{(2n-r-1)(n-s)}{(n+1)(2n+1-r)}, -\frac{(2n-r-1)(-r+n-1)}{(2n+1-r)(n-r+s)}\}$$

Koepf # 9.13

```
> rec:=sumrecursion(binomial(n,k)*pochhammer(c,k)*pochhamm
  er(m,n-k)*
  hyperterm([-k,a,b],[c,d],1,j),j,s(k));
```

$$rec := -(k+2)(m+n-k-1)(n-2+m-k)(k+d+1)\,\mathrm{s}(k+2)$$
$$+ (n-k)(n-1-k)(c+k)(a+b-k-c-d)\,\mathrm{s}(k) - (n-1-k)(m+n-k-1)$$
$$(-3k+bk-2k^2+a+ba-1-2kd-2d-cd-2c+ak+b-2kc)\,\mathrm{s}(k+1) =$$
$$0$$

```
> rec2hyper(rec,s(k));
```

$$\{\ \}$$

Koepf # 9.16

```
> read ‘qsum.mpl‘;
```

Copyright 1998, Harald Boeing & Wolfram Koepf

Konrad-Zuse-Zentrum Berlin

Rogers

```
> F:=(-1)^k*q^(k*(3*k-1)/2)/qpochhammer(q,q,n+k)/qpochhamm
  er(q,q,n-k):
  RE:=qsumrecursion(F,q,k,S(n));
```

$$RE := -(-q^{(2n)}+q)(q^n-1)\,q\,(1+q^n)\,\mathrm{S}(n)$$
$$+ (q^{(2n+2)}-q^2-q^4-q^3+q^{(3n)}+q^{(3n+1)})\,\mathrm{S}(n-1)$$
$$+ (q^2+q^4+q^3+q^{(2n)})\,q\,\mathrm{S}(n-2) - q^5\,\mathrm{S}(n-3) = 0$$

```
> qrecsolve(RE,q,S(n),return=qhypergeometric);
```

$$\left[\left[\frac{1}{\mathrm{qpochhammer}(q,q,n)}, 0 \le n\right]\right]$$

There is one q-hypergeometric solution (up to multiplicative constant). Call it S1. The recursion has 2 other linearly independent solutions, S2 and S3, that we have not discovered. The general solution is A1*S1+A2*S2+A3*S3 for constants (w.r.t. n) A1,A2,A3.
Check 3 initial conditions to confirm that it is just S1 (=1*S1+0*S2+0*S3).

```
> Fn := proc(n0)
        local k0;
        qsimpcomb(sum(subs(k=k0,n=n0,F),k0=-n0..n0))
  end:
  'Fn(n)'$'n'=0..5;
  'qsimplify(Fn(n)/(1/qpochhammer(q,q,n)))'$'n'=0..5;
```

$$1, -\frac{1}{-1+q}, \frac{1}{(-1+q)^2\,(q+1)}, -\frac{1}{(-1+q)^3\,(q+1)\,(q^2+q+1)},$$
$$\frac{1}{(-1+q)^4\,(q+1)^2\,(q^2+1)\,(q^2+q+1)},$$
$$-\frac{1}{(-1+q)^5\,(q+1)^2\,(q^4+q^3+q^2+q+1)\,(q^2+q+1)\,(q^2+1)}$$

$$1, 1, 1, 1, 1, 1$$

"Creative Symmetrizing"

```
> rat:=qsimpcomb(subs(k=-k,F)/F);
```

$$rat := q^k$$

```
> qsumrecursion((1+rat)/2*F,q,k,S(n));
```

$$(-q^n+1)\,\mathrm{S}(n) - \mathrm{S}(n-1) = 0$$

reduceorder(rec,f,n,u,w) -- implement the reduction of order formula given in the answer key

A=B p. 165 # 4

First do the whole problem for f(n). Part (a):

```
> F := (n,k) -> binomial(3*k,k) * binomial(3*n-3*k,n-k);
  sumF := proc(n)
       local k;
       sum(F(n,k),k=0..n)
  end;
```

$$F := (n, k) \to \mathrm{binomial}(3\,k, k)\,\mathrm{binomial}(3\,n-3\,k, n-k)$$

$$sumF := \mathbf{proc}(n)\ \mathbf{local}\ k;\ \mathrm{sum}(\mathrm{F}(n, k), k = 0\,..\,n)\ \mathbf{end}$$

```
> recf := sumrecursion(F(n,k),k,f(n));
```

$$recf := 8\,(n+2)\,(2\,n+3)\,\mathrm{f}(n+2) - 6\,(36\,n^2+99\,n+70)\,\mathrm{f}(n+1)$$
$$+ 81\,(3\,n+4)\,(3\,n+2)\,\mathrm{f}(n) = 0$$

(b)

```
> hyper_f := rechyper(recf,f(n));
```

$$hyper\_f := \{\frac{27}{4}\}$$

So the only hypergeometric solution is

```
> f1_sol := n -> (27/4)^n;
```

$$f1\_sol := n \to \left(\frac{27}{4}\right)^n$$

```
> recw := reduceorder(recf,f,n,f1_sol(n),w);
```

$$recw := 9\,(n+2)\,(2\,n+3)\,\mathrm{w}(n+1) - 2\,(3\,n+4)\,(3\,n+2)\,\mathrm{w}(n) = 0$$

This leads to a solution for w(n):

```
> w1_sol := unapply(rsolve(recw,w(n)),n);
```

$$w1\_sol := n \to 3\,\frac{\sqrt{3}\,\mathrm{w}(0)\,\Gamma\left(\frac{4}{3}+n\right)\Gamma\left(n+\frac{2}{3}\right)4^n}{\pi\,\Gamma(2\,n+3)}$$

Note the book has a simpler looking answer using binomial coefficients. Reconcile it:

```
> w2_sol := n -> binomial(3*(n+1),n+1)/(3*n+2) *
  (27/4)^(-n);
```

$$w2\_sol := n \to \frac{\mathrm{binomial}(3\,n+3,\,n+1)\left(\frac{27}{4}\right)^{(-n)}}{3\,n+2}$$

```
> simplify(w1_sol(n)/w2_sol(n));
```

$$\frac{2}{3}\,\mathrm{w}(0)$$

(We have n+1 for the book’s n because they did reduction of order with the backwards antidifference v(n)-v(n-1), while we used the forwards antidifference v(n+1)-v(n).) Thus, v(n) = v(0) + w(0)+w(1)+...+w(n-1); there is no chance whatsoever that this will be hypergeometric for any choice of the constant v(0), because otherwise, it would have been discovered already by Hyper.

```
> v1_sol := unapply(Sum(w2_sol(k),k=0..n-1),n):
  f2_sol := unapply(f1_sol(n)*v1_sol(n),n);
```

$$f2\_sol := n \to \left(\frac{27}{4}\right)^n \left(\sum_{k=0}^{n-1} \frac{\mathrm{binomial}(3\,k+3,\,k+1)\left(\frac{27}{4}\right)^{(-k)}}{3\,k+2}\right)$$

Now determine the constants so that f(n) = A*f1(n) + B*f2(n).

```
> evalf2 := n -> eval(subs(Sum=sum,f2_sol(n)));
```

$$evalf2 := n \to \mathrm{eval}(\mathrm{subs}(Sum = sum,\,\mathrm{f2\_sol}(n)))$$

```
> eqs := 'sumF(n)=A*f1_sol(n)+B*evalf2(n)'$'n'=0..1;
```

$$eqs := 1 = A,\ 6 = \frac{27}{4} A + \frac{81}{8} B$$

```
> solve({eqs},{A,B});
```

$$\{A = 1,\ B = \frac{-2}{27}\}$$

```
> full_f := subs(",A*f1_sol(n)+B*f2_sol(n));
```

$$full\_f := \left(\frac{27}{4}\right)^n - \frac{2}{27}\left(\frac{27}{4}\right)^n \left(\sum_{k=0}^{-1+n} \frac{\mathrm{binomial}(3\,k+3,\,k+1)\left(\frac{27}{4}\right)^{(-k)}}{3\,k+2}\right)$$

```
>
```

Now do this all for g(n). Part (a):

```
> G := (n,k)->binomial(3*k,k) * binomial(3*n-3*k-2,n-k-1);
  sumG := proc(n)
       local k;
       sum(G(n,k),k=0..n)
  end;
```

$$G := (n, k) \rightarrow \mathrm{binomial}(3\,k,\,k)\ \mathrm{binomial}(3\,n - 3\,k - 2,\, n - k - 1)$$

$$sumG := \mathbf{proc}(n)\ \mathbf{local}\ k;\ \mathrm{sum}(\mathrm{G}(n,k),\, k = 0\,..\,n)\ \mathbf{end}$$

```
> recg := sumrecursion(G(n,k),k,g(n));
```

$$recg := -16\,(2\,n+5)\,(2\,n+3)\,(n+2)\,\mathrm{g}(n+3)$$
$$+ 12\,(2\,n+3)\,(54\,n^2 + 153\,n + 130)\,\mathrm{g}(n+2)$$
$$- 324\,(27\,n^3 + 72\,n^2 + 76\,n + 30)\,\mathrm{g}(n+1) + 2187\,(3\,n+2)\,(3\,n+1)\,n\,\mathrm{g}(n) = 0$$

(b)

```
> hyper_g := rechyper(recg,g(n));
```

$$hyper\_g := \{\frac{27}{4},\ \frac{27}{2}\,\frac{n}{2\,n+1}\}$$

```
> ratio(27^n/(n*binomial(2*n,n)),n);
```

$$\frac{27}{2}\,\frac{n}{2\,n+1}$$

So there are two hypergeometric solutions.

```
> g1_sol := n -> (27/4)^n;
  # g2_sol := n -> (27/4)^n * GAMMA(n)/GAMMA(n+1/2); #
  straigtforward
  g2_sol := n->27^n/(n*binomial(2*n,n)); # simpler looking
```

$$g1\_sol := n \to \left(\frac{27}{4}\right)^n$$

$$g2\_sol := n \to \frac{27^n}{n \, \mathrm{binomial}(2\,n, n)}$$

but there is another linearly independent solution because the recursion has order 3. Reduce order to find it.

```
> recw1 := reduceorder(recg,g,n,g1(n),w1);
```

$$recw1 := 4\,n\,(3\,n+2)\,(3\,n+1)\,\mathrm{w1}(n) - 4\,(2\,n+3)\,(9\,n^2+18\,n+10)\,\mathrm{w1}(n+1) + 9\,(2\,n+5)\,(2\,n+3)\,(n+2)\,\mathrm{w1}(n+2) = 0$$

Now reduce the order again. We could either find a solution of recg_w from scratch, or we can use the second solution of the original recursion to create such a solution:

```
> dg2 := simpcomb(g2_sol(n+1)/g1_sol(n+1) -
  g2_sol(n)/g1_sol(n)):
  dg2 := unapply(dg2,n); # turn it into a function
```

$$dg2 := n \to -\frac{27^n\,\Gamma(n+1)\,\Gamma(n)}{\left(\frac{27}{4}\right)^n \Gamma(2\,n+2)}$$

Check it:

```
> simplify(eval(subs(w1=dg2,recw1)));
```

$$0 = 0$$

Cleaner solution:

```
> dg2b := n -> 4^n / (n*(2*n+1)*binomial(2*n,n));
  simplify(dg2(n)/dg2b(n));
```

$$dg2b := n \to \frac{4^n}{n\,(2\,n+1)\,\mathrm{binomial}(2\,n, n)}$$

$$-1$$

Do the second reduction of order:

```
> recw2 := reduceorder(recw1,w1,n,dg2b(n),w2);
```

$$recw2 := -(3\,n+2)\,(3\,n+1)\,\mathrm{w2}(n) + 9\,(n+1)\,(n+2)\,\mathrm{w2}(n+1) = 0$$

The automatic solution for this is

```
> w2_auto := rsolve(recw2,w2(n));
```

$$w2\_auto := \frac{1}{2}\,\frac{\Gamma\left(n+\frac{2}{3}\right)\Gamma\left(\frac{1}{3}+n\right)(n+1)\,\sqrt{3}\,\mathrm{w2}(0)}{\Gamma(n+2)^2\,\pi}$$

and a cleaner looking solution is

```
> w2_sol := n ->
```

```
  binomial(3*n,n)*binomial(2*n,n)/(n+1)/27^n;
```

$$w2\_sol := n \rightarrow \frac{\mathrm{binomial}(3\,n, n)\,\mathrm{binomial}(2\,n, n)}{(n+1)\,27^n}$$

Check it:

```
> simpcomb(eval(subs(w2=w2_sol,recw2)));
```

$$0 = 0$$

Then v2 is

```
> v2_sol := gosper(w2_sol(n),n);
```

$$v2\_sol := \frac{9}{2}\,\frac{n\,\mathrm{binomial}(3\,n, n)\,\mathrm{binomial}(2\,n, n)}{27^n}$$

So w1 is

```
> w1_sol := unapply(v2_sol * dg2b(n),n);
```

$$w1\_sol := n \rightarrow \frac{9}{2}\,\frac{\mathrm{binomial}(3\,n, n)\,4^n}{27^n\,(2\,n+1)}$$

Check:

```
> simpcomb(eval(subs(w1=w1_sol,recw1)));
```

$$0 = 0$$

We don't expect this to work, because the original recursion doesn't have 3 lin. ind. hypergeometric solutions:

```
> v1_sol := gosper(w1_sol(n),n);
Error, (in gosper) no hypergeometric term antidifference exists
```

So instead use

```
> v1_sol := Sum(w1_sol(k),k=0..n-1);
```

$$v1\_sol := \sum_{k=0}^{-1+n} \left(\frac{9}{2}\,\frac{\mathrm{binomial}(3\,k, k)\,4^k}{27^k\,(2\,k+1)} \right)$$

```
> g3_sol := unapply(g1_sol(n) * v1_sol,n);
```

$$g3\_sol := n \rightarrow \left(\frac{27}{4}\right)^n \left(\sum_{k=0}^{n-1} \left(\frac{9}{2}\,\frac{\mathrm{binomial}(3\,k, k)\,4^k}{27^k\,(2\,k+1)} \right) \right)$$

Now match initial conditions.

```
> eqs :=
  'sumG(n)=A*g1_sol(n)+B*g2_sol(n)+C*g3_sol(n)'$'n'=1..3:
> eqs := eval(subs(Sum=sum,{eqs}));
```

$$eqs := \{1 = \frac{27}{4}\,A + \frac{27}{2}\,B + \frac{243}{8}\,C,\ 48 = \frac{19683}{64}\,A + \frac{6561}{20}\,B + \frac{215055}{128}\,C,$$

$$7 = \frac{729}{16}\,A + \frac{243}{4}\,B + \frac{7533}{32}\,C\}$$

```
> sols := solve(eqs,{A,B,C}):
> g_full :=
  subs(sols,A*g1_sol(n)+B*g2_sol(n)+C*g3_sol(n));
```

$$g\_full := \frac{1}{9}\left(\frac{27}{4}\right)^n + \frac{2}{243}\left(\frac{27}{4}\right)^n \left(\sum_{k=0}^{-1+n} \left(\frac{9}{2}\,\frac{\mathrm{binomial}(3\,k,k)\,4^k}{27^k\,(2\,k+1)}\right)\right)$$

which is equivalent to the solution given in A=B, p. 167.

```
>
```

(c) the book has a complete answer.